NAG Library Routine Document

G05PGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05PGF generates a given number of terms of an exponential GARCH(p, q) process (see Engle and Ng (1993)).

2 Specification

```
SUBROUTINE G05PGF(DIST, NUM, IP, IQ, THETA, DF, HT, ET, FCALL, R, LR,
1                 STATE, IFAIL)
INTEGER           NUM, IP, IQ, DF, LR, STATE(lstate), IFAIL
double precision  THETA(2*IQ+IP+1), HT(NUM), ET(NUM), R(LR)
LOGICAL           FCALL
CHARACTER*1       DIST
```

3 Description

An exponential GARCH(p, q) process is represented by:

$$ln(h_t) = \alpha_0 + \sum_{i=1}^{q} \alpha_i z_{t-i} + \sum_{i=1}^{q} \phi_i(|z_{t-i}| - E[|z_{t-i}|]) + \sum_{j=1}^{p} \beta_j ln\big(h_{t-j}\big), \qquad t = 1, 2, \ldots, T;$$

where $z_t = \dfrac{\epsilon_t}{\sqrt{h_t}}$, $E[|z_{t-i}|]$ denotes the expected value of $|z_{t-i}|$, and $\epsilon_t \mid \psi_{t-1} = N(0, h_t)$ or $\epsilon_t \mid \psi_{t-1} = S_t(df, h_t)$. Here S_t is a standardized Student's t-distribution with df degrees of freedom and variance h_t, T is the number of observations in the sequence, ϵ_t is the observed value of the GARCH(p, q) process at time t, h_t is the conditional variance at time t, and ψ_t the set of all information up to time t.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05PGF.

4 References

Bollerslev T (1986) Generalised autoregressive conditional heteroskedasticity *Journal of Econometrics* **31** 307–327

Engle R (1982) Autoregressive conditional heteroskedasticity with estimates of the variance of United Kingdom inflation *Econometrica* **50** 987–1008

Engle R and Ng V (1993) Measuring and Testing the Impact of News on Volatility *Journal of Finance* **48** 1749–1777

Glosten L, Jagannathan R and Runkle D (1993) Relationship between the expected value and the volatility of nominal excess return on stocks *Journal of Finance* **48** 1779–1801

Hamilton J (1994) *Time Series Analysis* Princeton University Press

5 Parameters

1: DIST – CHARACTER*1 *Input*

On entry: the type of distribution to use for ϵ_t.

DIST = 'N'
A Normal distribution is used.

DIST = 'T'
A Student's t-distribution is used.

Constraint: DIST = 'N' or 'T'.

2: NUM – INTEGER *Input*

On entry: T, the number of terms in the sequence.

Constraint: NUM ≥ 0.

3: IP – INTEGER *Input*

On entry: the number of coefficients, β_i, for $i = 1, 2, \ldots, p$.

Constraint: IP ≥ 0.

4: IQ – INTEGER *Input*

On entry: the number of coefficients, α_i, for $i = 1, 2, \ldots, q$.

Constraint: IQ ≥ 1.

5: THETA($2 \times$ IQ + IP + 1) – ***double precision*** array *Input*

On entry: the initial parameter estimates for the vector θ. The first element must contain the coefficient α_o and the next IQ elements must contain the autoregressive coefficients α_i, for $i = 1, 2, \ldots, q$. The next IQ elements must contain the coefficients ϕ_i, for $i = 1, 2, \ldots, q$. The next IP elements must contain the moving average coefficients β_j, for $j = 1, 2, \ldots, p$.

Constraints:

$$\sum_{i=1}^{p} \beta_i \neq 0.0;$$
$$\frac{\alpha_0}{\sum_{i=1}^{p} \beta_i} \leq -\log(\text{X02AMF}).$$

6: DF – INTEGER *Input*

On entry: the number of degrees of freedom for the Student's t-distribution.

If DIST = 'N', DF is not referenced.

Constraint: if DIST = 'T', DF > 2.

7: HT(NUM) – ***double precision*** array *Output*

On exit: the conditional variances h_t, for $t = 1, 2, \ldots, T$, for the GARCH(p, q) sequence.

8: ET(NUM) – ***double precision*** array *Output*

On exit: the observations ϵ_t, for $t = 1, 2, \ldots, T$, for the GARCH(p, q) sequence.

9: FCALL – LOGICAL *Input*

On entry: if FCALL = .TRUE., a new sequence is to be generated, otherwise a given sequence is to be continued using the information in R.

10: R(LR) – ***double precision*** array *Input/Output*

On entry: the array contains information required to continue a sequence if FCALL = .FALSE..

On exit: contains information that can be used in a subsequent call of G05PGF, with FCALL = .FALSE..

11: LR – INTEGER *Input*

On entry: the dimension of the array R as declared in the (sub)program from which G05PGF is called.

Constraint: LR $\geq 2 \times (\text{IP} + 2 \times \text{IQ} + 2)$.

12: STATE($lstate$) – INTEGER array *Communication Array*

Note: $lstate$ = LSTATE, the dimension of STATE as supplied to the initialization routine G05KFF or G05KGF.

On entry: contains information on the selected base generator and its current state.

On exit: contains updated information on the state of the generator.

13: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, DIST $\neq$ 'N' or 'T'.

IFAIL = 2

On entry, NUM ≤ 0.

IFAIL = 3

On entry, IP < 0.

IFAIL = 4

On entry, IQ < 1.

IFAIL = 6

On entry, DIST = 'T' and DF ≤ 2.

IFAIL = 10

The value of IP or IQ is not the same as when R was set up in a previous call.

IFAIL = 11

On entry, LR $< 2 \times (\text{IP} + \text{IQ} + 2)$.

IFAIL = 12

On entry, STATE vector was not initialized or has been corrupted.

IFAIL = 20

Invalid sequence generated, use different parameters.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example first calls G05KFF to initialize a base generator then calls G05PGF to generate two realizations, each consisting of ten observations, from an exponential GARCH$(1, 1)$ model.

9.1 Program Text

```
*     G05PGF Example Program Text
*     Mark 22 Release. NAG Copyright 2006.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          MSTATE, MSEED, NUM, IP, IQ
      PARAMETER        (MSTATE=633,MSEED=1,NUM=10,IP=1,IQ=1)
      INTEGER          LR
      PARAMETER        (LR=2*(2*IQ+IP+2))
*     .. Local Scalars ..
      INTEGER          DF, GENID, I, IFAIL, LSEED, LSTATE, SUBID
      LOGICAL          FCALL
      CHARACTER        DIST
*     .. Local Arrays ..
      DOUBLE PRECISION ET(NUM), HT(NUM), R(LR), THETA(2*IP+IQ+1)
      INTEGER          SEED(MSEED), STATE(MSTATE)
*     .. External Subroutines ..
      EXTERNAL         G05KFF, G05PGF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05PGF Example Program Results'
      WRITE (NOUT,*)
*     Initialize the seed
      SEED(1) = 1762543
*     GENID and SUBID identify the base generator
      GENID = 1
      SUBID = 1
*     Initialize the generator to a repeatable sequence
      LSTATE = MSTATE
      LSEED = MSEED
      IFAIL = 1
      CALL G05KFF(GENID,SUBID,SEED,LSEED,STATE,LSTATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99998) IFAIL
         GO TO 60
      END IF
*     Set up the parameters for the series being generated
      DIST = 'N'
      THETA(1) = 0.1D0
      THETA(2) = -0.3D0
```

```
      THETA(3) = 0.1D0
      THETA(4) = 0.9D0
      DF = 0
      FCALL = .TRUE.
*     Generate the first realisation
      IFAIL = 1
      CALL G05PGF(DIST,NUM,IP,IQ,THETA,DF,HT,ET,FCALL,R,LR,STATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99997) IFAIL
         GO TO 60
      END IF
*     Display the results
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' Realisation Number 1'
      WRITE (NOUT,*) '   I            HT(I)            ET(I)'
      WRITE (NOUT,*) '  ---------------------------------------'
      DO 20 I = 1, NUM
         WRITE (NOUT,99999) I, HT(I), ET(I)
   20 CONTINUE
*     Generate a second realisation
      FCALL = .FALSE.
      IFAIL = 1
      CALL G05PGF(DIST,NUM,IP,IQ,THETA,DF,HT,ET,FCALL,R,LR,STATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99997) IFAIL
         GO TO 60
      END IF
*     Display the results
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' Realisation Number 2'
      WRITE (NOUT,*) '   I            HT(I)            ET(I)'
      WRITE (NOUT,*) '  ---------------------------------------'
      DO 40 I = 1, NUM
         WRITE (NOUT,99999) I, HT(I), ET(I)
   40 CONTINUE
*
   60 CONTINUE
*
99999 FORMAT (1X,I5,2(1X,F16.4))
99998 FORMAT (1X,' ** G05KFF returned with IFAIL = ',I5)
99997 FORMAT (1X,' ** G05PGF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05PGF Example Program Results


  Realisation Number 1
    I            HT(I)            ET(I)
   ---------------------------------------
     1           2.5098           0.5526
     2           2.1785          -1.8383
     3           3.3844           1.2180
     4           2.6780           1.3672
     5           2.0953          -1.8178
     6           3.2813          -0.0343
     7           2.9958          -0.5094
     8           3.0815           1.3978
     9           2.3961          -0.0070
    10           2.2445           0.6661

  Realisation Number 2
    I            HT(I)            ET(I)
   ---------------------------------------
     1           1.9327          -2.2795
```

```
     2          3.5577         -1.2249
     3          4.1461          0.6424
     4          3.4455         -2.9920
     5          5.9199          0.5777
     6          4.8221         -1.2894
     7          5.3174         -1.6473
     8          6.1095          6.1689
     9          3.1579          2.2935
    10          2.2189          0.1141
```